
· A subscreen area is a reserved rectangular area on a screen, in which you place another screen at run time. Subscreen areas may not contain any other screen elements. To use a subscreen, you create a second screen (with the type subscreen) and display it in the subscreen area you defined on the main screen. 

· A subscreen is an independent screen which you display within another screen. You may want to use a subscreen as a way of displaying a group of objects from the main screen in certain circumstances, but not in others. You can use this technique to display or hide extra fields on the main screen, depending on the entries the user has made.

· A second use for subscreens is that different programs can use the same subscreens. To set this up, you must execute other screen programs within your main program.

· You can include more than one subscreen on a single main screen. You can also determine the subscreens dynamically at run time. 

· You can use subscreens in the following circumstances:

·   In screen enhancements (screen exits)

·   Within other screen objects (tabstrip controls)

·   In the Modification Assistant

·   In Web transactions 

· If the subscreen is larger than the subscreen area in which it is called, the system displays only what will fit on the screen, starting at the upper-left corner. However, you can use the Scrollable attribute to ensure that, if the screen is too big, the system will display scrollbars.  

· The resizing attributes control whether the size of a subscreen area can be changed vertically and horizontally. You should set these attributes if you want the size of the subscreen area to change with the size of the whole window. You can use the minimum size attribute to set a lower limit beyond which the subscreen area cannot be resized. 

· The Context menu attribute allows you to assign a context-sensitive menu to the output fields on the subscreen. 

· The following restrictions apply to subscreens:

· CALL SUBSCREEN  is not allowed between LOOP and ENDLOOP or between 
CHAIN and ENDCHAIN.

· A subscreen may not have a named OK_CODE field. 

· Object names must be unique within the set of all subscreens called in a single main screen.

· Subscreens may not contain a module with the AT EXIT-COMMAND addition.
· You cannot use the SET TITLEBAR, SET PF-STATUS, SET SCREEN, or LEAVE SCREEN statements in the modules of a subscreen.

· To create a subscreen area, choose subscreen from the object list in the Screen Painter and place it on the screen. Fix the top-left corner of the table control area and then drag the object to the required size.

· In the Object text field, enter a name for the subscreen area. You need this to identify the area when you call the subscreen.

· To use a subscreen, you must call it in both the PBO and PAI sections of the flow logic of the main screen. The CALL SUBSCREEN <subarea> statement tells the system to execute the PBO and PAI processing blocks for the subscreen as components of the PBO and PAI of the main screen. You program the ABAP modules for subscreens in the same way as for a normal screen (apart from the restrictions already mentioned).

· The fields that you use within the flow logic are global fields of your ABAP program. These fields must be declared in the TOP include of your program.

· If the subscreen is not in the same module pool as the main program, the global data of the main program is not available to the subscreen and the data from the screen will not be transferred back to the program. You must program the data transfer yourself (for example, using a function module that exports and imports data, with an appropriate MOVE statement in the subscreen coding).  

· If you want to use subscreens in the screens of several different programs, you encapsulate the subscreens in a function group and use function modules to transport data between the program in which you want to use the subscreen and the function group. 

· You pass data between the calling program and the function group using the interfaces of the function modules.

· This is the technique used for customer subscreens (screen enhancements).

· You use function modules to transport data between the calling program and the function group.

· To declare the data from the calling program to the subscreen from the function group, use a module before the subscreen call. This calls a function module whose interface you can use to pass the required data to the function group.

· The function module call must occur before the subscreen call. This ensures that the data is known in the function group before the PROCESS BEFORE OUTPUT processing block of the subscreen is called.

· The sequence is reversed in the PAI module of the calling screen. You call the PROCESS AFTER INPUT processing block of the subscreen before you call a function module to pass the data from the function group back to the calling program.

· For the data from the calling program to be available globally in the function group, you must transfer the interface parameters from the function module into global data fields of the function group.

· The function module that you use to transfer the data from the calling program into the function group must copy its interface parameters into the global data in the function group.

· The function module that you use to transfer data from the function group to the calling program must copy the corresponding data from the global data of the function group into its interface parameters.

· Tabstrip controls provide you with an easy, user-friendly way of displaying different components of an application on a single screen and allowing the user to navigate between them. Their intuitive design makes navigation much easier for end users.  

· Tabstrip controls are a useful way of simplifying complex applications. You can use tabstrip controls wherever you have different components of an application that form a logical unit. For example, you might have a set of header data that remains constant, while underneath it  you want to display various other sets of data.

· You should not use tabstrip controls if:

· You need to change the screen environment (menus, pushbuttons, header data, and so on) while processing the application components. The screen surrounding the tabstrip must remain constant.

· The components must be processed in a certain order. Tabstrip controls are designed to allow users to navigate freely between components.

· The components are processed dynamically, that is, user input on one tab page causes other tab pages to suddenly appear. 

· Tabstrip controls are compatible with batch input processing. 

· A tabstrip control consists of individual pages with a tab page and tab title.

· The tabstrip control may have only one row of tab titles.

· If the tabstrip control contains too many pages, it is not possible for all of the tab titles to be displayed at once. In this case, a scrollbar allows you to scroll through the remaining tab pages. In the upper-right corner of the tab is a pushbutton. If the user selects this pushbutton, a list of all of the tab titles is displayed. The active tab title is marked with a checkmark.

· A page element consists of a tab title, a subscreen area, and a subscreen.

· From a technical point of view, the system handles tab titles like pushbuttons. 

· The contents of page elements are displayed using the subscreen technique. You assign a subscreen area to each page element for which you can then call a subscreen. 

· In addition to the general attributes Object name, Starting position, and static size, tabstrip controls also have special attributes.

· For details of these special attributes, see the section in this unit on subscreen attributes.

· You create a tabstrip control by carrying out the following three steps:

· 
Define the tab area.

· 
Define the tab titles and, if necessary, add further tab titles.

· 
Assign a subscreen area to each page element.

· To create a tabstrip control area, choose Tabstrip control from the object list in the Screen Painter and place it on the screen. Fix the upper-left corner of the table control area and then drag the object to the required size.

· Assign a name to the tabstrip control in the Object name attribute. You need this name to identify your tabstrip control.

· In your ABAP program, use the CONTROLS statement to declare an object with the same name. Use TABSTRIP as the type. 
· The type TABSTRIP is defined in the type pool CXTAB. The ACTIVETAB field contains the function code of the tab title of the currently active tabstrip. The other fields are reserved for internal use.

· The default number of tab pages for a tabstrip control is two.

· Technically, tab titles are treated in the same way as pushbuttons. They have a name, a text, a function code, and a function type. You enter these in the Name, Text, FctCode, and FctType fields of the object attributes.

· A tab title can have the function type ' ' (space) or P. If the function type is ' ' (space), the PAI processing block is triggered when the user chooses that tab and the function code of the tab title is placed in the command field. If the function type is P, the user can scroll between different tab pages of the same type without triggering the PAI processing block. If you want your tabstrip control to have more than two pages, you must create further tab titles. To do this, choose Pushbutton from the object list in the Screen Painter and place it in the tab title area.

· You must assign a subscreen area to each tab page.

· The subscreen area assigned to a tab page is automatically entered as the Reference object (in the Dictionary attributes) for the tab title of that page. 

· To assign a subscreen area to one or more tab pages, choose the relevant tab title in the fullscreen editor, choose the Subscreen object, and place it on the tab page.

· Alternatively, you can assign a single subscreen area to several tab pages by entering the name of the subscreen area directly in the Reference object field of the attributes of the relevant tab pages.

· If you have assigned a different subscreen area to each page element in a tabstrip control, you can scroll between the pages locally at the front end.

· To do this, you must send all of the subscreens to the front end when you send the main screen itself. All of the tab titles in the tabstrip control must also have function type P.

· When you scroll between the different page elements, there is no communication between the presentation server and the application server.

· When the user chooses a function on the screen that triggers PAI processing, the system processes the PAI blocks of all of the subscreens as well. This means that all of the field checks are run. In this respect, you could regard the tabstrip control as behaving like a single large screen. 

· Local scrolling in tabstrip controls is more appropriate for display transactions. 

· To program a tabstrip control to scroll locally at the front end, you must:

· Assign a separate subscreen area to each tab page. A subscreen will be sent to each of these when the screen is processed.

· Call all of the subscreens from the flow logic.

· Assign function type P to all of the tab titles.

· The system hides any page element whose subscreen contains no elements that can be displayed.

· If there are no page elements containing elements that can be displayed, the system hides the entire tabstrip control.

· For more information about tabstrip controls, refer to the online documentation, SUB-1.

· If all of the page elements share a single subscreen area, the program analyzes the function code of the chosen tab title to determine which screen is displayed.

· There are two steps in this process:

· In the PAI processing block, the program determines which page element needs to be active, based on the tab title chosen by the user.

· When the PBO processing block is processed again, the program displays the corresponding screen.

· During this process, the system checks only the fields of the displayed subscreen. 

· If you want the application program to process scrolling in a tabstrip control, the following requirements must be met:

· All of the tab pages must share a common subscreen area.

· All of the tab titles must have the function code type ' ' (space).

· In the flow logic, you must use a variable to call the screen that is to be displayed in the subscreen area.

· In the PAI block, you must call a module in which the function code of the active tab title is placed in the ACTIVETAB field of the structure you created in your program with type TABSTRIP. In the example in the graphic, this is MY_TAB_STRIP.
· The PBO processing block must contain a module before the subscreen is called, in which you place the name of the subscreen in the corresponding variable. You must assign an initial value to this field so that the screen is processed the first time (before the user has had a chance to choose a tab title).

· You can hide a tab page at run time by setting the corresponding tab title to inactive using the system table SCREEN (SCREEN-ACTIVE = 0). You should do this before processing the tabstrip control for the first time to ensure that the screen environment remains constant.

· You can use the Tabstrip Control Wizard to help you create tabstrip controls and insert them on screens in a program. The Wizard guides you through the process. You can return to previous settings at any time. Program objects are created upon the final screen only on completion of the process. The Wizard creates not only the tabstrip control but also the corresponding statements in the flow logic, together with the relevant modules, subroutines, and necessary data definitions. 

· In addition to the tabstrip control on the screen and the corresponding flow logic, the following objects are created if they do not already exist:

· The main program and the screen for the tabstrip control together with its flow logic.

· Empty subscreens for the individual tabstrip control pages.

· Includes for data definition, PBO modules, PAI modules, and INCLUDE statements for these includes.

· All objects are placed in the inactive object list.

Subscreens and Tabstrip Controls Exercises

	Unit: Subscreens and Tabstrip Controls

Topic: Creating subscreens and tabstrip controls

	At the conclusion of these exercises, you will be able to:

· Use subscreens and tabstrip controls on screens and selection screens in your programs.

	Display additional information on your screen, depending on the mode in which the user is working. 

Extend the display to allow users to switch between the additional information using a tabstrip control. 


1-1
Extend the Maintenance screen (100) to display flight information and the aircraft type. Use a subscreen to do this.  

1-1-1
Extend your program SAPMZ##BC410_SOLUTION from the previous exercise (or copy the model solution SAPMBC410AINPS_RADIOBUTTON). You can use the model solution SAPMBC410ASUBS_SUBSCREEN for orientation. 
1-1-2
On the maintenance screen (100), create a subscreen area with the following attributes:

	Subscreen
	SUB
	Attributes: 
Vertical and horizontal Resizing:  
ON


1-1-3
Create three screens 110, 120, and 130, each with the type subscreen and the following attributes: 

	Screen 110
	I/O fields, text fields: 


SDYN_CONN
   - COUNTRYFR
   - COUNTRYTO
   - CITYFROM
   - CITYTO
   - AIRPFROM
   - AIRPTO
   - DEPTIME
   - ARRTIME
	For each field:


 Input: OFF
Output: ON



	Screen 120
	I/O Fields, Text fields: 
SAPLANE
   - PLANETYPE
   - PRODUCER
   - SEATSMAX
   - TANKCAP
   - CAP_UNIT
   - WEIGHT
   - WEI_UNIT
   - OP_SPEED
   - SPEED_UNIT
	Attributes for each field: 
    Input: OFF
    Output: ON
    Output only: ON

	Screen 130
	empty (Provided for the bookings table)
	


1-1-4
In your TOP include, create a field DYNNR that you can use in the flow logic to determine which subscreen should appear in the subscreen area.

1-1-5
Call the subscreen screens in the flow logic of screen 100. Before the call, write a PBO module to determine which of the subscreens will appear.  If the user is in Display mode, call subscreen screen 110 with the flight information. If the user is in Maintain flight data mode, call subscreen screen 120 with the aircraft information. If the user chooses Maintain bookings mode, then empty screen 130 appears.

1-1-6
In the flow logic of screen 110, read the flight information from table SPFLI using the key field values. 

1-1-7
In the flow logic for screen 120, read the information for the aircraft information from table SAPLANE using the value you have for the aircraft type.  

1-2
Create a tabstrip control on screen 100 for displaying extra flight information and details of the aircraft type.  

1-2-1
Extend your program SAPMZ##BC410_SOLUTION from the previous exercise (or copy the model solution SAPMBC410ASUBS_SUBSCREEN). You can use the model solution SAPMBC410ASUBS_TABSTRIP for orientation. 
1-2-2
Creating a tabstrip: Remove the subscreen area on screen 100 and create a tabstrip control with the following attributes: 

	Tabstrip controls
	Name:
MY_TABSTRIP
	Attributes: 
Vertical and horizontal Resizing:  ON

	Pushbutton
(Tab title 1)
	Name:
P1
	Attributes:
Text: View flight data
Function code: FC1
Function type: <blank>
Reference field: SUB

	Pushbutton
(Tab title 2)
	Name:
P2
	Attributes:
Text: View technical data for aircraft
Function code: FC2
Function type: <blank>
Reference field: SUB

	Pushbutton
(Tab title 3)
	Name:
P3
	Attributes:
Text: Maintain booking
Function code: FC3
Function type: <blank>
Reference field: SUB


In the TOP include of your program, create a data object for the tabstrip control using the following statement: 
CONTROLS MY_TABSTRIP ...
1-2-3
In the flow logic of screen 100, implement the call for the subscreen screen in the tabstrip control. The subscreen number will be in the field DYNNR created in the previous exercise.

1-2-4
Before calling the subscreen, write a PBO module in which you determine which of the subscreens is to be called (regardless of the mode in which the user is working). Additionally, determine which subscreen screen you want to set the first time the screen is displayed and assign the corresponding function code to the field MY_TABSTRIP-ACTIVETAB.

1-2-5
Extend your command field processing for screen 100 with the scroll logic for the tabstrip control. Do this by assigning the relevant value to MY_TABSTRIP-ACTIVETAB.

Subscreens and Tabstrip Controls Solutions

	Unit: Subscreens and Tabstrip Controls

Topic: Creating subscreens and tabstrip controls


1-1
Model solution SAPMBC410ASUBS_SUBSCREEN
Add the coding in bold type, and create new modules where appropriate using forward navigation.

Top include

PROGRAM  sapmbc410asubs_subscreen MESSAGE-ID bc410         .

DATA  dynnr TYPE sy-dynnr.                                  "#EC NEEDED
TABLES saplane.

TABLES sdyn_conn.

DATA: BEGIN OF mode,

        view VALUE 'X',                "selected

        maintain_flights,

        maintain_bookings,

      END OF mode.

DATA ok_code TYPE sy-ucomm.

DATA wa_sflight TYPE sflight.

Subroutine include

No changes are necessary. 

Flow logic screen 100

PROCESS BEFORE OUTPUT.

  MODULE status_0100.

  MODULE modify_screen.

  MODULE fill_dynnr.

  CALL SUBSCREEN sub INCLUDING sy-cprog dynnr.

  MODULE clear_ok_code.

*

PROCESS AFTER INPUT.

  MODULE exit AT EXIT-COMMAND.
*  CALL SUBSCREEN sub.

  CHAIN.

    FIELD: sdyn_conn-carrid,

           sdyn_conn-connid,

           sdyn_conn-fldate  MODULE check_sflight ON CHAIN-REQUEST.

  ENDCHAIN.

  CHAIN.

    FIELD: sdyn_conn-planetype,

           sdyn_conn-seatsmax MODULE check_planetype ON CHAIN-REQUEST.

  ENDCHAIN.

  MODULE trans_from_dynp.

  MODULE user_command_0100.

Flow logic screen 110

PROCESS BEFORE OUTPUT.

  MODULE get_spfli.

PROCESS AFTER INPUT.

Flow logic screen 120

PROCESS BEFORE OUTPUT.

  MODULE get_saplane.

PROCESS AFTER INPUT.

Flow logic screen 130

No changes are necessary. 

PBO module include

MODULE status_0100 OUTPUT.

  SET PF-STATUS 'STATUS_100'.

  SET TITLEBAR 'TITLE_100'.

ENDMODULE.                             " STATUS_0100  OUTPUT
MODULE clear_ok_code OUTPUT.

  CLEAR ok_code.

ENDMODULE.                             " clear_ok_code  OUTPUT
MODULE modify_screen OUTPUT.

  CHECK NOT mode-maintain_flights IS INITIAL.

  LOOP AT SCREEN.

    IF screen-name = 'SDYN_CONN-PLANETYPE'.

      screen-input = 1.


 screen-required = 1.

 MODIFY SCREEN.

    ENDIF.

  ENDLOOP.

ENDMODULE.                             " modify_screen  OUTPUT
MODULE get_spfli OUTPUT.

  ON CHANGE OF wa_sflight-carrid

            OR wa_sflight-connid.

    SELECT SINGLE * INTO CORRESPONDING FIELDS OF sdyn_conn FROM spfli

      WHERE carrid = wa_sflight-carrid

        AND connid = wa_sflight-connid.

  ENDON.

ENDMODULE.                             " GET_SPFLI  OUTPUT
MODULE get_saplane OUTPUT.

  ON CHANGE OF wa_sflight-planetype.

    SELECT SINGLE * FROM saplane

 WHERE planetype = wa_sflight-planetype.

  ENDON.

ENDMODULE.                             " GET_SAPLANE  OUTPUT
MODULE fill_dynnr OUTPUT.

  CASE 'X'.

    WHEN mode-view.

      dynnr = 110.

    WHEN mode-maintain_flights.

      dynnr = 120.

    WHEN mode-maintain_bookings.

      dynnr = 130.

  ENDCASE.

ENDMODULE.                             " set_dynnr  OUTPUT
PAI module include

MODULE user_command_0100 INPUT.

  CASE ok_code.

    WHEN 'SAVE'.

      PERFORM update_sflight.

    WHEN 'BACK'.

      LEAVE TO SCREEN 0.

  ENDCASE.

ENDMODULE.                             " USER_COMMAND_0100  INPUT

MODULE trans_from_dynp INPUT.

  MOVE-CORRESPONDING sdyn_conn TO wa_sflight.

ENDMODULE.                             " trans_from_dynp  INPUT

MODULE read_sflight INPUT.

  SELECT SINGLE * INTO CORRESPONDING FIELDS OF sdyn_conn FROM sflight

    WHERE carrid = sdyn_conn-carrid

      AND connid = sdyn_conn-connid

      AND fldate = sdyn_conn-fldate.

  IF sy-subrc NE 0.

    MESSAGE e038.

  ENDIF.

ENDMODULE.                             " read_sflight  INPUT

MODULE exit INPUT.

  CASE ok_code.

    WHEN 'EXIT'.

      LEAVE PROGRAM.

    WHEN 'CANCEL'.

      CLEAR: sdyn_conn, saplane, wa_sflight.

      SET PARAMETER ID: 'CAR' FIELD space,

                        'CON' FIELD space,

                        'DAY' FIELD space.

      LEAVE TO SCREEN 100.

  ENDCASE.

ENDMODULE.                             " exit  INPUT

MODULE check_planetype INPUT.

  SELECT SINGLE seatsmax INTO sdyn_conn-seatsmax FROM saplane

    WHERE planetype = sdyn_conn-planetype.

  CHECK sdyn_conn-seatsmax < sdyn_conn-seatsocc.

  MESSAGE e109.

ENDMODULE.                             " check_planetype  INPUT
1-2
Model solution SAPMBC410ASUBS_TABSTRIP

Add the coding in bold type, and create new modules where appropriate using forward navigation.

Top include

CONTROLS my_tabstrip TYPE TABSTRIP.

Subroutine include

No changes are necessary. 

Flow logic screen 100

No changes are necessary. 

Flow logic screen 110

No changes are necessary. 

Flow logic screen 120

No changes are necessary. 

Flow logic screen 130

No changes are necessary. 

PBO module include

MODULE status_0100 OUTPUT.

  SET PF-STATUS 'STATUS_100'.

  SET TITLEBAR 'TITLE_100'.

ENDMODULE.                             " STATUS_0100  OUTPUT

MODULE clear_ok_code OUTPUT.

  CLEAR ok_code.

ENDMODULE.                             " clear_ok_code  OUTPUT

MODULE modify_screen OUTPUT.

  CHECK NOT mode-maintain_flights IS INITIAL.

  LOOP AT SCREEN.

    IF screen-name = 'SDYN_CONN-PLANETYPE'.

      screen-input = 1.

      screen-required = 1.

      MODIFY SCREEN.

    ENDIF.

  ENDLOOP.

ENDMODULE.                             " modify_screen  OUTPUT

MODULE get_spfli OUTPUT.

  ON CHANGE OF wa_sflight-carrid OR wa_sflight-connid.

    SELECT SINGLE * INTO CORRESPONDING FIELDS OF sdyn_conn FROM spfli

      WHERE carrid = wa_sflight-carrid

        AND connid = wa_sflight-connid.

  ENDON.

ENDMODULE.                             " GET_SPFLI  OUTPUT

MODULE get_saplane OUTPUT.

  ON CHANGE OF wa_sflight-planetype.

    SELECT SINGLE * FROM saplane

      WHERE planetype = wa_sflight-planetype.

  ENDON.

ENDMODULE.                             " GET_SAPLANE  OUTPUT

MODULE fill_dynnr OUTPUT.

  CASE my_tabstrip-activetab.

    WHEN 'FC1'.

      dynnr = 110.

    WHEN 'FC2'.

      dynnr = 120.

    WHEN 'FC3'.

      dynnr = 130.

    WHEN OTHERS.

      my_tabstrip-activetab = 'FC1'.

      dynnr = 110.

  ENDCASE.
PAI module include

MODULE user_command_0100 INPUT.

  CASE ok_code.

     WHEN 'FC1' or 'FC2' or 'FC3'.

      my_tabstrip-activetab = ok_code.
    WHEN 'SAVE'.

      PERFORM update_sflight.

    WHEN 'BACK'.

      LEAVE TO SCREEN 0.

  ENDCASE.

ENDMODULE.                             " USER_COMMAND_0100  INPUT
MODULE trans_from_dynp INPUT.

  MOVE-CORRESPONDING sdyn_conn TO wa_sflight.

ENDMODULE.                             " trans_from_dynp  INPUT

MODULE read_sflight INPUT.

  SELECT SINGLE * INTO CORRESPONDING FIELDS OF sdyn_conn FROM sflight

    WHERE carrid = sdyn_conn-carrid

      AND connid = sdyn_conn-connid

      AND fldate = sdyn_conn-fldate.

  IF sy-subrc NE 0.

    MESSAGE e038.

  ENDIF.

ENDMODULE.                             " read_sflight  INPUT

MODULE exit INPUT.

  CASE ok_code.

    WHEN 'EXIT'.

      LEAVE PROGRAM.

    WHEN 'CANCEL'.

      CLEAR: sdyn_conn, saplane, wa_sflight.

      SET PARAMETER ID: 'CAR' FIELD space,

                        'CON' FIELD space,

                        'DAY' FIELD space.

      LEAVE TO SCREEN 100.

  ENDCASE.

ENDMODULE.                             " exit  INPUT

MODULE check_planetype INPUT.

  SELECT SINGLE seatsmax INTO sdyn_conn-seatsmax FROM saplane

    WHERE planetype = sdyn_conn-planetype.

  CHECK sdyn_conn-seatsmax < sdyn_conn-seatsocc.

  MESSAGE e109.

ENDMODULE.                             " check_planetype  INPUT

